

### EXAMPLE 1.9-1: Temperature Sensor Error due to Mounting & Self Heating

A resistance temperature detector (RTD) utilizes a material that has an electrical resistivity that is a strong function of temperature. The temperature of the RTD is inferred by measuring its electrical resistance. Figure 1 shows an RTD that is mounted at the end of a metal rod and inserted into a pipe in order to measure the temperature of a flowing liquid. The RTD is monitored by passing a known current through it and measuring the voltage across it. This process results in a constant amount of ohmic heating that may tend to cause the RTD temperature to rise relative to the temperature of the surrounding liquid; this effect is referred to as a self-heating measurement error. Also, conduction from the wall of the pipe to the temperature sensor through the metal rod can result in a temperature difference between the RTD and the liquid; this effect is referred to as a mounting measurement error.




Figure 1: Temperature sensor mounted in a flowing liquid.

The thermal energy generation associated with ohmic heating is  $\dot{q}_{sh} = 2.5$  mW. All of this ohmic heating is assumed to be transferred from the RTD into the end of the rod at  $x = L$ . The rod has a thermal conductivity  $k = 10$  W/m-K, diameter  $D = 0.5$  mm, and length  $L = 5.0$  cm. The end of the rod that is connected to the pipe wall (at  $x = 0$ ) is maintained at a temperature of  $T_w = 20^\circ\text{C}$ .

The liquid is at a uniform temperature,  $T_\infty = 5^\circ\text{C}$ . However, the local heat transfer coefficient between the liquid and the rod ( $h$ ) varies with  $x$  due to the variation of the liquid velocity in the pipe. This problem resembles external flow over a cylinder, which will be discussed in Chapter 4; however, you may assume that the heat transfer coefficient between the rod surface and the fluid varies according to:

$$h = 2000 \left[ \frac{\text{W}}{\text{m}^{2.8} \text{K}} \right] x^{0.8} \quad (1)$$

where  $h$  is the heat transfer coefficient in  $\text{W}/\text{m}^2\text{-K}$  and  $x$  is position along the rod in m.

a.) Can the rod be treated as an extended surface?

The input parameters are entered in EES; note that the heat transfer coefficient is computed using a function defined at the top of the EES code.

"EXAMPLE 1.9-1: Temperature Sensor Error due to Mounting and Self Heating"

\$UnitSystem SI MASS DEG PA C J

\$Tabstops 0.2 0.4 0.6 0.8 3.5

"Function for heat transfer coefficient"

```
function h(x)
  h=2000 [W/m^2.8-K]*x^0.8
end
```

"Inputs"

q\_dot\_sh=2.5 [milliW]\*convert(milliW,W)

"self-heating power"

k=10 [W/m-K]

"conductivity of mounting rod"

D=0.5 [mm]\*convert(mm,m)

"diameter of mounting rod"

L=5.0 [cm]\*convert(cm,m)

"length of mounting rod"

T\_w=converttemp(C,K,20 [C])

"temperature of wall"

T\_infinity=converttemp(C,K,5 [C])

"temperature of liquid"

The appropriate Biot number for this case is:

$$Bi = \frac{hD}{2k}$$

The Biot number will be largest (and therefore the extended surface approximation least valid) when the heat transfer coefficient is largest. According to Eq. (1), the highest heat transfer coefficient occurs at the tip of the rod; therefore, the Biot number is calculated according to:

Bi=h(L)\*D/(2\*k)

"Biot number"

The Biot number calculated by EES is 0.0046, which is much less than 1.0 and therefore the extended surface approximation is justified.

b.) Develop a numerical model of the rod that will predict the temperature distribution in the rod and therefore the error in the temperature measurement; this error is the difference between the temperature at the tip of the rod (i.e., the temperature of the RTD) and the liquid.

The development of the numerical model follows the same steps that are discussed in Section 1.4. Nodes (i.e., locations where the temperature will be determined) are positioned uniformly along the length of the rod, as shown in Figure 2. The location of each node ( $x_i$ ) is:

$$x_i = \frac{(i-1)}{(N-1)}L \quad i = 1..N$$

where  $N$  is the number of nodes used for the simulation. The distance between adjacent nodes ( $\Delta x$ ) is:

$$\Delta x = \frac{L}{(N-1)}$$


This distribution is entered in EES:

```

N=100                                     "number of nodes"
duplicate i=1,N
  x[i]=(i-1)*L/(N-1)                     "position of each node"
end
DELTAx=L/(N-1)                           "distance between adjacent nodes"

```

A control volume is defined around each node; the control surface bisects the distance between the nodes, as shown in Figure 2.



**Figure 2: Control volume for an internal node.**

The control volume for the internal node  $i$  shown in Figure 2 is subject to conduction heat transfer at each edge ( $\dot{q}_{top}$  and  $\dot{q}_{bottom}$ ) and convection ( $\dot{q}_{conv}$ ). The energy balance is:

$$\dot{q}_{top} + \dot{q}_{bottom} + \dot{q}_{conv} = 0$$

The conduction terms are approximated according to:

$$\dot{q}_{top} = \frac{k \pi D^2}{4 \Delta x} (T_{i-1} - T_i)$$

$$\dot{q}_{bottom} = \frac{k \pi D^2}{4 \Delta x} (T_{i+1} - T_i)$$

The convection term is modeled using the convection coefficient evaluated at the position of the node:

$$\dot{q}_{conv} = h_{x_i} \pi D \Delta x (T_\infty - T_i)$$

Combining these equations leads to:

$$\frac{k \pi D^2}{4 \Delta x} (T_{i-1} - T_i) + \frac{k \pi D^2}{4 \Delta x} (T_{i+1} - T_i) + h_{x_i} \pi D \Delta x (T_\infty - T_i) = 0 \quad \text{for } i = 2..(N-1) \quad (2)$$

"internal control volume energy balances"

```
duplicate i=2,(N-1)
  k*pi*D^2*(T[i-1]-T[i])/(4*DELTAx)+k*pi*D^2*(T[i+1]-T[i])/(4*DELTAx)+&
  pi*D*DELTAx*h(x[i])*(T_infinity-T[i])=0
end
```

The nodes at the edges of the domain must be treated separately. At the pipe wall, the temperature is specified:

$$T_1 = T_w \quad (3)$$

```
T[1]=T_w
```

"boundary condition at wall"

The ohmic dissipation,  $\dot{q}_{sh}$  is assumed to enter the half-node at the tip (i.e., node  $N$ ) and therefore is included in the energy balance for this node (see Figure 2):

$$\frac{k \pi D^2}{4 \Delta x} (T_{N-1} - T_N) + \frac{h_{x_N} \pi D \Delta x}{2} (T_\infty - T_N) + \dot{q}_{sh} = 0 \quad (4)$$


Note the factor of 2 in the denominator of the convection term that arises because the half-node has half the surface area of the internal nodes.

```
k*pi*D^2*(T[N-1]-T[N])/(4*DELTAx)+pi*D*DELTAx*h(x[N])*(T_infinity-T[N])/2+q_dot_sh=0
"boundary condition at tip"
```

Equations (2) through (4) are a system of  $N$  equations in an equal number of unknown temperatures which are entered in EES. The solution is converted to Celsius:

```
duplicate i=1,N
  T_C[i]=converttemp(K,C,T[i])
end
"solution in Celsius"
```

Figure 3 illustrates the temperature distribution in the rod for  $N = 100$  nodes. The temperature elevation of the tip relative to the fluid is about 3.4 K and represents the measurement error. For the conditions in the problem statement, it is clear that the measurement error is primarily due to self-heating because the effect of the wall (the temperature elevation at the base) has died off after about 2.0 cm.



**Figure 3: Temperature distribution in the mounting rod.**


As with any numerical solution, it is important to verify that a sufficient number of nodes have been used so that the numerical solution has converged. The key result of the solution is the tip-to-fluid temperature difference, or measurement error for the sensor ( $\delta T$ ):

$$\delta T = T_N - T_\infty$$

delta T = T[N] - T\_infinity

"measurement error"

Figure 4 illustrates the tip-to-fluid temperature difference as a function of the number of nodes and shows that the solution has converged to within 0.01 K for  $N$  greater than 100 nodes.



**Figure 4: Tip-to-fluid temperature difference as a function of the number of nodes.**

The analytical solution for this problem in the limit of a constant heat transfer coefficient and an adiabatic tip was derived in Section 1.6.3 and included in Table 1-4:

$$\frac{T - T_\infty}{T_w - T_\infty} = \frac{\cosh(m(L-x))}{\cosh(mL)}$$

where

$$m = \sqrt{\frac{4\bar{h}}{kD}}$$

The analytical solution is programmed in EES:


```
"Analytical solution for verification in the limit q_dot_sh=0 and h=constant"
m=sqrt(4*h(L)/(k*D)) "fin parameter"
duplicate i=1,N
  T_an[i]=T_infinity+(T_w-T_infinity)*cosh(m*(L-x[i]))/cosh(m*L) "analytical solution"
  T_an_C[i]=converttemp(K,C,T_an[i]) "in C"
end
```

The EES program is run in this limit by setting the variable  $q_{dot\_sh} = 0$  and modifying the function  $h$  so that it returns  $100 \text{ W/m}^2\text{-K}$  regardless of position.

```
"Function for heat transfer coefficient"
function h(x)
  {h=2000 [W/m^2.8-K]*x^0.8}
  h=100 [W/m^2-K]
end
```

```
"Inputs"
q_dot_sh=0 [W] {2.5 [milliW]*convert(milliW,W)} "self-heating power"
```

The temperatures predicted by the numerical model are compared with the analytical solution in Figure 5.




**Figure 5: Verification of the numerical model against the analytical solution in the limit that the heat transfer coefficient is constant at  $h = 100 \text{ W/m}^2\text{-K}$  and there is no self-heating,  $\dot{q}_{sh} = 0 \text{ W}$ .**

c.) Investigate the effect of thermal conductivity on the temperature measurement error. Identify the optimal thermal conductivity and explain why an optimal thermal conductivity exists.

Figure 6 illustrates the temperature measurement error as a function of the thermal conductivity of the rod material; note that the function  $h$  has been set back to its original form and the variable  $q_{dot\_sh}$  restored to  $2.5 \text{ mW}$ . Figure 6 shows that the optimal thermal conductivity, corresponding to the minimum measurement error, is around  $100 \text{ W/m-K}$ . Below the optimal

value, the self-heating error dominates as the local temperature rise at the tip of the rod is large. Above the optimal value, the conduction from the wall dominates.



**Figure 6: Temperature measurement error as a function of rod thermal conductivity.**